F07VSF (CTBTRS/ZTBTRS) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F07VSF (CTBTRS/ZTBTRS) solves a complex triangular band system of linear equations with multiple right-hand sides, $AX = B$, $A^T X = B$ or $A^H X = B$.

2 Specification

```
 SUBROUTINE F07VSF(UPLO, TRANS, DIAG, N, KD, NRHS, AB, LDAB, B,
1                  LDB, INFO)
 ENTRY      ctbtrs(UPLO, TRANS, DIAG, N, KD, NRHS, AB, LDAB, B,
1                  LDB, INFO)
 INTEGER           N, KD, NRHS, LDAB, LDB, INFO
 complex           AB(LDAB,*), B(LDB,*)
 CHARACTER*1       UPLO, TRANS, DIAG
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine solves a complex triangular band system of linear equations $AX = B$, $A^T X = B$ or $A^H X = B$.

4 References

[1] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

[2] Higham N J (1989) The accuracy of solutions to triangular systems *SIAM J. Numer. Anal.* **26** 1252–1265

5 Parameters

1: UPLO — CHARACTER*1 *Input*

On entry: indicates whether A is upper or lower triangular as follows:

if UPLO = 'U', then A is upper triangular;
if UPLO = 'L', then A is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: TRANS — CHARACTER*1 *Input*

On entry: indicates the form of the equations as follows:

if TRANS = 'N', then the equations are of the form $AX = B$;
if TRANS = 'T', then the equations are of the form $A^T X = B$;
if TRANS = 'C', then the equations are of the form $A^H X = B$.

Constraint: TRANS = 'N', 'T' or 'C'.

3: DIAG — CHARACTER*1 *Input*

On entry: indicates whether A is a non-unit or unit triangular matrix as follows:

if DIAG = 'N', then A is a non-unit triangular matrix;
if DIAG = 'U', then A is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.

Constraint: DIAG = 'N' or 'U'.

4: N — INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

5: KD — INTEGER *Input*

On entry: k, the number of super-diagonals of the matrix A if UPLO = 'U' or the number of sub-diagonals if UPLO = 'L'.

Constraint: KD ≥ 0.

6: NRHS — INTEGER *Input*

On entry: r, the number of right-hand sides.

Constraint: NRHS ≥ 0.

7: AB(LDAB,$*$) — ***complex*** array *Input*

Note: the second dimension of the array AB must be at least max(1,N).

On entry: the n by n triangular band matrix A, stored in rows 1 to $(k+1)$. More precisely, if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element a_{ij} in AB$(k+1+i-j,j)$ for $\max(1,j-k) \leq i \leq j$;if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element a_{ij} in AB$(1+i-j,j)$ for $j \leq i \leq \min(n,j+k)$. If DIAG = 'U', the diagonal elements of A are not referenced and are assumed to be 1.

8: LDAB — INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07VSF (CTBTRS/ZTBTRS) is called.

Constraint: LDAB $\geq$ KD + 1.

9: B(LDB,$*$) — ***complex*** array *Input/Output*

Note: the second dimension of the array B must be at least max(1,NRHS) .

On entry: the n by r right-hand side matrix B.

On exit: the n by r solution matrix X.

10: LDB — INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07VSF (CTBTRS/ZTBTRS) is called.

Constraint: LDB $\geq$ max(1,N).

11: INFO — INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, a_{ii} is zero and the matrix A is singular.

7 Accuracy

The solutions of triangular systems of equations are usually computed to high accuracy. See Higham [2].

For each right-hand side vector b, the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

$$|E| \leq c(k)\epsilon|A|,$$

$c(k)$ is a modest linear function of k, and ϵ is the ***machine precision***.

If $\hat{x}$ is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_\infty}{\|x\|_\infty} \leq c(k)\text{cond}(A, x)\epsilon, \quad \text{provided} \quad c(k)\text{cond}(A, x)\epsilon < 1,$$

where $\text{cond}(A, x) = \| \, |A^{-1}||A||x| \, \|_\infty / \|x\|_\infty$.

Note that $\text{cond}(A, x) \leq \text{cond}(A) = \| \, |A^{-1}||A| \, \|_\infty \leq \kappa_\infty(A)$; $< \text{cond}(A, x)$ can be much smaller than $< \text{cond}(A)$ and it is also possible for $\text{cond}(A^H)$, which is the same as $\text{cond}(A^T)$, to be much larger (or smaller) than $\text{cond}(A)$.

Forward and backward error bounds can be computed by calling F07VVF (CTBRFS/ZTBRFS), and an estimate for $\kappa_\infty(A)$ can be obtained by calling F07VUF (CTBCON/ZTBCON) with NORM = 'I'.

8 Further Comments

The total number of real floating-point operations is approximately $8nkr$ if $k \ll n$.

The real analogue of this routine is F07VEF (STBTRS/DTBTRS).

9 Example

To solve the system of equations $AX = B$, where

$$A = \begin{pmatrix} -1.94 + 4.43i & 0.00 + 0.00i & 0.00 + 0.00i & 0.00 + 0.00i \\ -3.39 + 3.44i & 4.12 - 4.27i & 0.00 + 0.00i & 0.00 + 0.00i \\ 1.62 + 3.68i & -1.84 + 5.53i & 0.43 - 2.66i & 0.00 + 0.00i \\ 0.00 + 0.00i & -2.77 - 1.93i & 1.74 - 0.04i & 0.44 + 0.10i \end{pmatrix}$$

and

$$B = \begin{pmatrix} -8.86 - 3.88i & -24.09 - 5.27i \\ -15.57 - 23.41i & -57.97 + 8.14i \\ -7.63 + 22.78i & 19.09 - 29.51i \\ -14.74 - 2.40i & 19.17 + 21.33i \end{pmatrix}.$$

Here A is treated as a lower triangular band matrix with 2 sub-diagonals.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07VSF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KDMAX, LDAB, NRHMAX, LDB
      PARAMETER        (NMAX=8,KDMAX=NMAX,LDAB=KDMAX+1,NRHMAX=NMAX,
     +                 LDB=NMAX)
      CHARACTER        TRANS, DIAG
      PARAMETER        (TRANS='N',DIAG='N')
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, KD, N, NRHS
      CHARACTER        UPLO
*     .. Local Arrays ..
      complex          AB(LDAB,NMAX), B(LDB,NRHMAX)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Subroutines ..
      EXTERNAL         ctbtrs, X04DBF
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07VSF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD, NRHS
      IF (N.LE.NMAX .AND. KD.LE.KDMAX .AND. NRHS.LE.NRHMAX) THEN
*
*        Read A and B from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            DO 20 I = 1, N
               READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))
   20       CONTINUE
         ELSE IF (UPLO.EQ.'L') THEN
            DO 40 I = 1, N
               READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)
   40       CONTINUE
         END IF
         READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*        Compute solution
*
         CALL ctbtrs(UPLO,TRANS,DIAG,N,KD,NRHS,AB,LDAB,B,LDB,INFO)
*
*        Print solution
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
            IFAIL = 0
            CALL X04DBF('General',' ',N,NRHS,B,LDB,'Bracketed','F7.4',
     +                  'Solution(s)','Integer',RLABS,'Integer',CLABS,
     +                  80,0,IFAIL)
         ELSE
```

```
            WRITE (NOUT,*) 'A is singular'
          END IF
        END IF
        STOP
*
        END
```

9.2 Program Data

```
F07VSF Example Program Data
  4  2  2                                                  :Values of N, KD and NRHS
  'L'                                                       :Value of UPLO
 (-1.94, 4.43)
 (-3.39, 3.44) ( 4.12,-4.27)
 ( 1.62, 3.68) (-1.84, 5.53) ( 0.43,-2.66)
               (-2.77,-1.93) ( 1.74,-0.04) ( 0.44, 0.10)  :End of matrix A
 ( -8.86, -3.88) (-24.09, -5.27)
 (-15.57,-23.41) (-57.97,  8.14)
 ( -7.63, 22.78) ( 19.09,-29.51)
 (-14.74, -2.40) ( 19.17, 21.33)                           :End of matrix B
```

9.3 Program Results

```
 F07VSF Example Program Results

 Solution(s)
                     1                  2
 1  ( 0.0000, 2.0000) ( 1.0000, 5.0000)
 2  ( 1.0000,-3.0000) (-7.0000,-2.0000)
 3  (-4.0000,-5.0000) ( 3.0000, 4.0000)
 4  ( 2.0000,-1.0000) (-6.0000,-9.0000)
```